ABOUT THE BOOK

This book provides a detailed introduction to the methods used in experimental research for the reduction and error analysis of physical data. It gives a thorough treatment of techniques for fitting data by the method of least squares, starting at a level appropriate for use with a slide rule or mechanical calculator and continuing through sophisticated techniques appropriate for use with computers. The scope is that of undergraduate and graduate university study. No previous background of statistical concepts or techniques is required.

The basic approach is that of an introductory text, with a practical emphasis on the techniques and procedures to be followed but with enough derivation to justify the results. Typical experiments are described in each section and worked out with detailed calculations to illustrate the appropriate methods. Each chapter contains exercises and summaries of the definitions and formulas discussed. Fortran computer routines, written explicitly for each section, are included to illustrate and explain the use of the techniques discussed. The primary purpose of the routines is to show, by example, how typical calculations are carried out, but they are also usable in their entirety as library routines.

ABOUT THE AUTHOR

Philip R. Bevington is Associate Professor of Physics at Case Western Reserve University. He graduated from Harvard University in 1954 and received his Ph.D. from Duke University in 1960. He taught at Duke University for five years and was an Assistant Professor at Stanford University from 1963 to 1968 before coming to Case Western Reserve University.

The author has been engaged in research in nuclear structure physics with Van de Graaff accelerators. While at Stanford he was active in computer applications for nuclear physics and was responsible for development of the SCANS (Stanford Computers for the Analysis of Nuclear Structure) system. He is a member of the American Physical Society and Meetings Chairman of the Digital Equipment Computer Users Society.

McGraw-Hill Book Company
330 West 42nd Street
New York 10036
San Francisco • St. Louis
Toronto • London • Sydney

05135

Data Reduction and Error Analysis for the Physical Sciences



Philip R. Bevington
Associate Professor of Physics · Case Western Reserve University

and consider this to be the appropriate measure of the goodness of fit. We have used the same symbol χ^2 defined earlier in Equation (5-19) because this is essentially the same definition in a different context.

Our method for finding the optimum fit to the data will be to minimize this weighted sum of squares of deviations χ^2 and, hence, to find the fit which produces the smallest sum of squares or the *least-squares fit*.

6-3 INSTRUMENTAL UNCERTAINTIES

If the quantity y is one which can be measured with a physical instrument, the uncertainty in each measurement generally comes from fluctuations in repeated readings of the instrumental scale, either because the settings are not exactly reproducible due to imperfections in the equipment, or because of human imprecision in observing the settings, or a combination of both. Such uncertainties are called *instrumental* because they arise from a lack of perfect precision in the measuring instruments (including the observer).

We can include in this category experiments which deal with measurements of such characteristics as length, mass, voltage, current, etc. In the discussion which follows, we shall consider first the simpler case where the absolute uncertainties are equal throughout the entire experiment. Later we shall consider the refinement of utilizing the standard deviation as a weighting factor corresponding to the precision which may vary from one part of the experiment to another as when the scale factor is changed or the scale is non-linear. In the next section we will consider uncertainties resulting from statistical fluctuations rather than from experimental precision.

For example, we include in this category such experiments as that of Example 6-1 illustrated in Figure 6-1 in which the observed quantity is the temperature T, measured with a thermometer consisting of a thermocouple and a meter with a linear scale. The fluctuations in the data result from errors in reading the meter,

and these errors are just as large for readings near the low end of the scale as for readings near the high end (ignoring errors in calibration). So long as we do not change the scale to measure temperatures outside the reasonable range, the absolute values (rather than the relative values) of the uncertainties will be the same for all measurements.

Minimizing χ^2 In order to find the values of the coefficients a and b which yield the minimum value for χ^2, we use the method of calculus described in Appendix A in the same way as in Section 5-1, extrapolated to minimizing the function with respect to more than one coefficient. The minimum value of the function χ^2 of Equation (6-6) is one which yields a value of 0 for both of the partial derivatives with respect to each of the coefficients

$$
\begin{aligned}
\frac{\partial}{\partial a}\chi^2 &= \frac{\partial}{\partial a}\left[\frac{1}{\sigma^2}\Sigma(y_i - a - bx_i)^2\right] \\
&= \frac{-2}{\sigma^2}\Sigma(y_i - a - bx_i) = 0 \\
\frac{\partial}{\partial b}\chi^2 &= \frac{\partial}{\partial b}\left[\frac{1}{\sigma^2}\Sigma(y_i - a - bx_i)^2\right] \\
&= \frac{-2}{\sigma^2}\Sigma[x_i(y_i - a - bx_i)] = 0
\end{aligned}
\qquad (6\text{-}7)
$$

where we have for the present considered all of the standard deviations equal $\sigma_i = \sigma$.

These equations can be rearranged to yield a pair of simultaneous equations

$$
\begin{aligned}
\Sigma y_i &= \Sigma a + \Sigma b x_i = aN + b\Sigma x_i \\
\Sigma x_i y_i &= \Sigma a x_i + \Sigma b x_i^2 = a\Sigma x_i + b\Sigma x_i^2
\end{aligned}
\qquad (6\text{-}8)
$$

where we have substituted N for $\Sigma(1)$ since the sum runs for $i = 1$ to N. This development is discussed more fully in Appendix A.

We wish to solve Equations (6-8) for the coefficients a and b. This will give us the values of the coefficients for which χ^2, the

Program 6-1 LINFIT Least-squares fit to a straight line.

```
C        SUBROUTINE LINFIT
C
C        PURPOSE
C          MAKE A LEAST-SQUARES FIT TO DATA WITH A STRAIGHT LINE
C             Y = A + B*X
C
C        USAGE
C          CALL LINFIT (X, Y, SIGMAY, NPTS, MODE, A, SIGMAA, B, SIGMAB, R)
C
C        DESCRIPTION OF PARAMETERS
C          X       - ARRAY OF DATA POINTS FOR INDEPENDENT VARIABLE
C          Y       - ARRAY OF DATA POINTS FOR DEPENDENT VARIABLE
C          SIGMAY  - ARRAY OF STANDARD DEVIATIONS FOR Y DATA POINTS
C          NPTS    - NUMBER OF PAIRS OF DATA POINTS
C          MODE    - DETERMINES METHOD OF WEIGHTING LEAST-SQUARES FIT
C                    +1 (INSTRUMENTAL) WEIGHT(I) = 1./SIGMAY(I)**2
C                     0 (NO WEIGHTING) WEIGHT(I) = 1.
C                    -1 (STATISTICAL)  WEIGHT(I) = 1./Y(I)
C          A       - Y INTERCEPT OF FITTED STRAIGHT LINE
C          SIGMAA  - STANDARD DEVIATION OF A
C          B       - SLOPE OF FITTED STRAIGHT LINE
C          SIGMAB  - STANDARD DEVIATION OF B
C          R       - LINEAR CORRELATION COEFFICIENT
C
C        SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C          NONE
C
C        MODIFICATIONS FOR FORTRAN II
C          OMIT DOUBLE PRECISION SPECIFICATIONS
C          CHANGE DSQRT TO SQRTF IN STATEMENTS 67, 68, AND 71
C
```

sum of squares of the deviations of the data points from the calculated fit, is a minimum. The solution can be found in any one of a number of different ways, but, for generality for later similar but more complex situations, let us use the method of determinants. Appendix B contains a discussion of this method and gives the rules for obtaining a solution for any number of simultaneous equations.

The solutions are:

$$a = \frac{1}{\Delta}\begin{vmatrix} \Sigma y_i & \Sigma x_i \\ \Sigma x_i y_i & \Sigma x_i^2 \end{vmatrix} = \frac{1}{\Delta}(\Sigma x_i^2 \Sigma y_i - \Sigma x_i \Sigma x_i y_i)$$

$$b = \frac{1}{\Delta}\begin{vmatrix} N & \Sigma y_i \\ \Sigma x_i & \Sigma x_i y_i \end{vmatrix} = \frac{1}{\Delta}(N \Sigma x_i y_i - \Sigma x_i \Sigma y_i) \tag{6-9}$$

$$\Delta = \begin{vmatrix} N & \Sigma x_i \\ \Sigma x_i & \Sigma x_i^2 \end{vmatrix} = N \Sigma x_i^2 - (\Sigma x_i)^2$$

Table 6-1 shows a sample calculation for the data of our first

Program 6-1 LINFIT *(continued)*

```
      SUBROUTINE LINFIT (X,Y,SIGMAY,NPTS,MODE,A,SIGMAA,B,SIGMAB,R)
      DOUBLE PRECISION SUM, SUMX, SUMY, SUMX2, SUMXY, SUMY2
      DOUBLE PRECISION XI, YI, WEIGHT, DELTA, VARNCE
      DIMENSION X(1), Y(1), SIGMAY(1)
C
C        ACCUMULATE WEIGHTED SUMS
C
   11 SUM = 0.
      SUMX = 0.
      SUMY = 0.
      SUMX2 = 0.
      SUMXY = 0.
      SUMY2 = 0.
   21 DO 50 I=1, NPTS
      XI = X(I)
      YI = Y(I)
      IF (MODE) 31, 36, 38
   31 IF (YI) 34, 36, 32
   32 WEIGHT = 1. / YI
      GO TO 41
   34 WEIGHT = 1. / (-YI)
      GO TO 41
   36 WEIGHT = 1.
      GO TO 41
   38 WEIGHT = 1. / SIGMAY(I)**2
   41 SUM   = SUM   + WEIGHT
      SUMX  = SUMX  + WEIGHT*XI
      SUMY  = SUMY  + WEIGHT*YI
      SUMX2 = SUMX2 + WEIGHT*XI*XI
      SUMXY = SUMXY + WEIGHT*XI*YI
      SUMY2 = SUMY2 + WEIGHT*YI*YI
   50 CONTINUE
C
C        CALCULATE COEFFICIENTS AND STANDARD DEVIATIONS
C
   51 DELTA = SUM*SUMX2 - SUMX*SUMX
      A = (SUMX2*SUMY - SUMX*SUMXY) / DELTA
   53 B = (SUMXY*SUM  - SUMX*SUMY ) / DELTA
   61 IF (MODE) 62, 64, 62
   62 VARNCE = 1.
      GO TO 67
   64 C = NPTS - 2
      VARNCE = (SUMY2 + A*A*SUM + B*B*SUMX2
     1 -2.*(A*SUMY + B*SUMXY - A*B*SUMX)) / C
   67 SIGMAA = DSQRT(VARNCE*SUMX2 / DELTA)
   68 SIGMAB = DSQRT(VARNCE*SUM   / DELTA)
   71 R = (SUM*SUMXY - SUMX*SUMY) /
     1 DSQRT(DELTA*(SUM*SUMY2 - SUMY*SUMY))
      RETURN
      END
```

example. The calculation is straightforward, though tedious. We accumulate four sums (Σx_i, $\Sigma y_i = \Sigma T_i$, Σx_i^2, and $\Sigma x_i y_i = \Sigma x_i T_i$), and combine them according to Equations (6-9) to find numerical values for a and b.

Program 6-1 The same method of calculation is also illustrated with the computer routine LINFIT of Program 6-1.

This is a Fortran subroutine to calculate the coefficients a and b for a least-squares fit of a straight line to an array of data points for any one of three different experimental conditions. The input variables are X, Y, SIGMAY, NPTS, and MODE, and the output variables are A, SIGMAA, B, SIGMAB, and R.

For the calculation discussed above, following the method of Equations (6-9), the variable MODE must have the value 0. This indicates to the subroutine that we have not considered any weighting of the fitting procedure by including the standard deviations of individual points. The variable NPTS represents the number of pairs of data points NPTS $= N$. The independent quantities x_i are assumed to be stored in the array X, and the dependent data points are assumed to be stored in the array Y, with the ordering identical for the two arrays. The array SIGMAY may be ignored; in this mode the subroutine does not use it or modify it.

The four sums given above (Σx_i, Σy_i, Σx_i^2, and $\Sigma x_i y_i$) are accumulated in statements 41–50 as part of the DO loop starting at statement 21. The variable WEIGHT is given a value of 1 in statement 36 and can be ignored. The calculations of Equations (6-9) are carried out in statements 51–53, with N replaced by SUM $= \Sigma(1)$. The coefficients A $= a$ and B $= b$ are returned to the calling program as arguments of the calling sequence. The remainder of the subroutine pertains to material not yet discussed.

Weighting the fit If the fluctuations in the data are due to instrumental errors, but for reasons of scale changes, nonlinear scales, etc., the uncertainties are not equal throughout, it is necessary to reintroduce the standard deviation from Equation (6-6) as a weighting factor into Equations (6-7) to (6-9). Instead of minimizing the simple sum of the squares of deviations as in Equations (6-7), we weight each term of the sum in χ^2 according to how large or small the deviation is expected to be at that point before summing.

Minimizing χ^2 as given in Equation (6-6), Equation (6-7)

becomes

$$\begin{aligned}\frac{\partial}{\partial a}\chi^2 &= \frac{\partial}{\partial a}\sum\left[\frac{1}{\sigma_i^2}(y_i - a - bx_i)^2\right]\\ &= -2\sum\left[\frac{1}{\sigma_i^2}(y_i - a - bx_i)\right] = 0\\ \frac{\partial}{\partial b}\chi^2 &= \frac{\partial}{\partial b}\sum\left[\frac{1}{\sigma_i^2}(y_i - a - bx_i)^2\right]\\ &= -2\sum\left[\frac{x_i}{\sigma_i^2}(y_i - a - bx_i)\right] = 0\end{aligned} \tag{6-10}$$

These equations can be rearranged to yield a pair of simultaneous equations analogous to Equations (6-8).

$$\begin{aligned}\sum\frac{y_i}{\sigma_i^2} &= a\sum\frac{1}{\sigma_i^2} + b\sum\frac{x_i}{\sigma_i^2}\\ \sum\frac{x_i y_i}{\sigma_i^2} &= a\sum\frac{x_i}{\sigma_i^2} + b\sum\frac{x_i^2}{\sigma_i^2}\end{aligned} \tag{6-11}$$

The solutions are similar to Equations (6-9).

$$\begin{aligned}a &= \frac{1}{\Delta}\begin{vmatrix}\sum\frac{y_i}{\sigma_i^2} & \sum\frac{x_i}{\sigma_i^2}\\ \sum\frac{x_i y_i}{\sigma_i^2} & \sum\frac{x_i^2}{\sigma_i^2}\end{vmatrix} = \frac{1}{\Delta}\left(\sum\frac{x_i^2}{\sigma_i^2}\sum\frac{y_i}{\sigma_i^2} - \sum\frac{x_i}{\sigma_i^2}\sum\frac{x_i y_i}{\sigma_i^2}\right)\\ b &= \frac{1}{\Delta}\begin{vmatrix}\sum\frac{1}{\sigma_i^2} & \sum\frac{y_i}{\sigma_i^2}\\ \sum\frac{x_i}{\sigma_i^2} & \sum\frac{x_i y_i}{\sigma_i^2}\end{vmatrix}\\ &= \frac{1}{\Delta}\left(\sum\frac{1}{\sigma_i^2}\sum\frac{x_i y_i}{\sigma_i^2} - \sum\frac{x_i}{\sigma_i^2}\sum\frac{y_i}{\sigma_i^2}\right)\\ \Delta &= \begin{vmatrix}\sum\frac{1}{\sigma_i^2} & \sum\frac{x_i}{\sigma_i^2}\\ \sum\frac{x_i}{\sigma_i^2} & \sum\frac{x_i^2}{\sigma_i^2}\end{vmatrix} = \sum\frac{1}{\sigma_i^2}\sum\frac{x_i^2}{\sigma_i^2} - \left(\sum\frac{x_i}{\sigma_i^2}\right)^2\end{aligned} \tag{6-12}$$

Such a calculation is even more tedious than that of Equations (6-9) and presupposes a knowledge of the magnitudes of the standard deviations σ_i for each of the data points. Fortunately, we are generally only interested in the relative uncertainties of various sections of our data when we modify scale factors, etc. For such purposes, the standard deviations can have any arbitrary overall normalization (e.g., the smallest value of σ_i^2 may be set equal to 1 so that all the other values are integers).

The method of calculation for the general case when the values of σ_i are known is illustrated in the subroutine LINFIT with the variable MODE given a value of +1 (or any positive integer). The standard deviations σ_i must be stored in the array SIGMAY with the same ordering as the data in arrays X and Y. The calculation is the same as for the earlier example, except that the variable WEIGHT is given the value 1. / (SIGMAY(I))^2 in statement 38 for each term so that the sums accumulated are those required for Equations (6-12).

6-4 STATISTICAL FLUCTUATIONS

If the quantity y represents the number of counts in a detector per unit time interval, as in Example 6-2, then it is generally true that the uncertainty in each measurement y_i is directly related to the magnitude of y (as discussed in Section 3-2), and, therefore, the standard deviations σ_i associated with these measurements cannot be considered equal over any reasonable range of values. Such uncertainties are called *statistical* because they arise not from a lack of perfect precision in the measuring instruments, but from statistical fluctuations in the collections of finite numbers of counts over finitely long intervals of time.

In our counting experiment of Example 6-2, for example, we would expect from the straight-line fit to the data that we should receive about 100 counts in our detector during the first time interval. What we mean by this is that if the counting rate were continued indefinitely at the same rate for a large number of intervals, the average number of counts received per interval would be very nearly 100. Since the counts are distributed randomly in time, however, we would expect to receive more than 100 counts in some intervals and fewer than 100 in others. The fluctuations

of the number of counts actually received in each interval around the average number of counts are statistical fluctuations related to the probability of receiving more or fewer than the average number of counts in any time interval.

There can be instrumental uncertainties as well contributing to the overall uncertainties. We can determine the time intervals with only finite precision, and the same precision applies to the determination of the starting times t_i, though this is generally a negligible correction. These are actually uncertainties in the independent quantity x, but we have agreed to assign them arbitrarily to the dependent quantity y. For counting experiments these contributions to the overall uncertainty are generally ignored on the assumption that the statistical fluctuations dominate. Where this is not true, the standard deviations to be used in Equations (6-10) to (6-12) as weighting factors must be the root sum squares of the standard deviations for the experimental deviations $\sigma_i(x_i)$ in x and the statistical deviations $\sigma_i(y_i)$ in y as given in Equation (6-1).

$$\sigma_i^2 = \sigma_i^2(x_i) + \sigma_i^2(y_i)$$

Estimate of σ If the fluctuations in the measurements y_i are statistical, we can estimate analytically what the standard deviation corresponding to each observation is, without having to determine it experimentally. If we were to make the same measurement repeatedly, we would find that the observed values were distributed about their mean in a Poisson distribution (as discussed in Section 3-2) instead of a Gaussian distribution. We can justify the use of this distribution intuitively by considering that we would expect a distribution which is related to the binomial distribution, but which is consistent with our boundary conditions that we may receive any positive number of counts, but no fewer than 0 counts, in any time interval.

One immediate advantage of the Poisson distribution is that the standard deviation is automatically determined.

$$\sigma = \sqrt{y} \tag{6-13}$$